
*                                                                       _______
*          ^  VCC                         _____________________________|       |
*         /_\                            |         ____________________|       |
*   _______|____            ^ VCC        |         |   RN2      ^ VCC  |       |
*   |          |           /_\      RN1  |         |           /_\     | BDITA |
*    ->      <-             |_/\  /\  /\_|         |_/\  /\  /\_|      |       |
*   QP4 `|_|' QP3               \/  \/   |VN1   VN2|   \/  \/          |       |
*     _ '|||`_                          _|         |_                  |       |
*    |____|   |                     __|' QN1     QN2 `|___             |       |
*    / VP4B   |                    |  |`->         <-'|   |            |       |
*     \       | VP4C               |     |_________|      |            |       |
*    / RS     |_______   _________/|\__       |  VNPN     |            |DUAL   |VIOUT     CCOMP
*    \VP5B  <-        | |   ___    |   \      |           |       ___  |       |   ___||_______
*    |____|'QP5     <-  |  |INP|___|    \     |           |______|INN| |INPUT  |   |  ||       |
*   / VP4 |`_    _|'QP6 |  |___|   |     \    |           |      |___| |       |   |           |
*    \        |  ||`____|   _|_    |      \   |           |         |  |TURN   |   |           |
*   / RS2     |  |         /VIN\   |       |  |           |         |  |       |___|_|\  _  _  |  ___
*   \    RS3  |  |         \___/   |     _/|\_|           |         |  |AROUND |     | \/ \/ \_|_|OUT|
*   |__/\   _/|\_|          _|_    |    /  |              |         |  |       |  ___| /\_/\_/   |___|
*   |    \/   | VP6B        ///    |   /   |     VPNP     |         |  |       |  |  |/           |
*   |VIPNP    |        ___________/|\_/   _|_______       |         |  |       |  |     BOTA      |
*  _|_        |        |           |     |         |      |         |  |       | _|_              |
* / _ \      V|4B      |           |   <-           ->    |         |  |       | \ /  VREF        |
* \/ \/       |___     |           |_|' QP1        QP2 `|_|         |  |       |  V               |
* /\_/\      |_   |   _|             |`_             _ '|           |  |       |                  |
* \___/        `|_|_|' QN4  _/\  /\  /\_|           |_/\  /\  /\_   |  |       |                  |
*  _|_       <-'|   |`->  _|_  \/  \/   |VP1     VP2|   \/  \/  _|_ |  |       |                  |
*  ///      _|_       _|_ \\\  RP1      |           |     RP2   /// |  |       |                  |
* IPNP      ///       ///               |           |______________/|\_|       |                  |
*            QN3                        |__________________________/|\_|       |                  |
*                                                                   |  |_______|                  |
*                                                                   |_____________________________|
*
*
.OPTIONS  method=gear  GMIN=1e-18

VIN       INP    0      SIN(6   6      1K )   
QP1       VP1    INP    VPNP    PNPV   1.04
QP2       VP2    OUT    VPNP    PNPV   1
QP3       VP4C   VP     VCC     PNPV   1
QP4       VP     VP     VCC     PNPV   1
RP1       VP1    0       5K
RP2       VP2    0       5K

QN1       VN1   INP     VNPN    NPNV   1.02
QN2       VN2   OUT     VNPN    NPNV   1
QN3       VNPN  VN      0       NPNV   1
QN4       VN    VN      0       NPNV   1



RN1       VCC   VN1     5K
RN2       VCC   VN2     5K

VCC       VCC    0      12 
VREF      VREF   0      6    
IPNP      VIPNP  0      20u
RS        VP     VP5B   10k
RS2       VP5B   VIPNP  10k
RS3       VIPNP  VP6B   150k
QP5       VN     VP5B   VP4C     PNPV   1
QP6       VPNP   VP6B   VP4C     PNPV   1

BDITA     VIOUT  0     I = ( V(VP2) - V(VP1) + V(VN2) - V(VN1)  )/5000
BOTA      OUT    0     I = -1*( V(VIOUT) - V(VREF) )/5000
CCOMP     OUT    VIOUT 3p

.model      NPNV npn BF=150

.model      PNPV pnp BF=150

.model      PNPL pnp BF=5

.control
run
set        pensize = 2
dc         vin  0    12     100m
plot       v(out)-v(inp) title  OffsetVoltageVsCommonMode
plot       v(vp1)   v(vp2)  v(vcc)-v(vn1)  v(vcc)-v(vn2) title  TailCurrentsVsCommonMode
*plot       v(vpnp)-v(vp5b) title  PnpEmitterBase
.endc
.end

* ====================END================================

The steering can be done using vertical pnps in the 
circuit above which allows the transitions to happen 
smoothly enough for a full rail to rail voltage follower
applications. 



This version of the steering does have a little
lower current gain for the npns. 



In the case of the LM6142, this gain difference was
preferred. This meant the part was more stable at
the top rail. Experience seems to suggest that this
may have kept a lot of customers out of trouble. 


